Use of a spreadsheet in the design of an industrial ventilation system

CATHERINE PRESTON KOSHLAND and MICHAEL G. YOST
Biomedical and Environmental Health Sciences, School of Public Health, University of California, Berkeley, California

Designing ventilation systems involves considerable computational effort which may be reduced via the use of a microcomputer. This article demonstrates how, by replacing the traditional row-column worksheet, a spreadsheet program may be utilized easily to perform ventilation system design. The article contains the equations and correlations necessary to define the mathematical relationships used to design a ventilation system. The equations and correlations are in forms which may be directly applied to most spreadsheet programs and may be used to develop a general ventilation template. Once this general ventilation template has been developed, it may be quickly adapted to any specific ventilation design problem.

The approach taken is to use the "velocity pressure" method of design. However, the basic concepts are applicable to use of the "equivalent foot" method. The equations are designed to minimize the manual use of psychrometric charts and tabular or graphical data. A design approach is outlined. How a spreadsheet functions is described. A sample design problem is discussed; the system layout, audit sheet, and ventilation template are included.

The use of a spreadsheet to perform ventilation system design calculations was validated in two studies. First, the results predicted by the spreadsheet model were compared to hand-calculated results given in textbook examples: on a scatter-plot, the points fell on a diagonal line of unit slope with the average error in the predicted values of −0.5 percent. Second, predicted static pressure values were compared to actual duct static pressure measurements: the average error in the predicted values was plus four percent. **Koshland, C.P.; Yost, M.G.:** Use of a spreadsheet in the design of an industrial ventilation system. *Appl. Ind. Hyg. 2:*204–212; 1987.

Introduction

Designing ventilation systems involves significant computational effort, but with the advent of microcomputers, much of the burden of calculation may be removed. Use of a microcomputer can greatly reduce the work, while providing greater flexibility, accuracy, and sophistication in the data analysis. The key element in using any microcomputer effectively is finding the software to accomplish the desired task.

Ventilation design programs specific to particular machines have been developed.[1–3] An alternative approach is to adapt a more general purpose program, such as a spreadsheet program, to design ventilation systems.

Spreadsheet programs were developed for tabular data analysis; the flexible format of the spreadsheet enables it to be used for a wide variety of industrial hygiene problems that utilize a row-column worksheet to organize the analysis. Users fill in the rows and columns with values and formulae that interconnect the entries. This method allows the user to build a template or spreadsheet model in which all the mathematical relationships for his specific task are user-defined. The template can be saved to disk and later recalled to perform the same calculation on another set of data.

Ventilation system design employing either the velocity pressure method or the equivalent foot method utilizes a row-column worksheet which is easily adapted to a spreadsheet. Each column of the spreadsheet constitutes the characteristics of one branch or section of duct; each row specifies a system characteristic. Use of the spreadsheet allows the design or analysis to be easily extended or refined. An initially simple design can be made more complex; a simple analysis can be refined with increased accuracy. The spreadsheet concept enables the user to perform "what if" calculations by varying the values of key parameters.

Since spreadsheet programs are available for virtually all microcomputers, the design method described in this paper is

TABLE I
Operating data inputs

T_{DB}	Dry-bulb temperature, C (F)
P_{baro}	Barometric pressure, mm Hg (in. Hg)
For each duct branch:	
Q_{dry}	Dry air volume flow standard, m³/s (scfm)
w	Humidity ratio, kg H_2O/kg air (lbs H_2O/lbs air)

0882-8032/87/0204-09$2.50



Figure 1—Ventilation design.

not machine-specific. Rather, users develop the spreadsheet model or "template" by providing data and formulae to an available spreadsheet program. Once the general ventilation template has been created, it may be adapted easily and quickly to almost any specific design problem. The initial development and testing of this spreadsheet template for ventilation design computations was performed on an Apple IIe microcomputer system with 128K of RAM and two disk drives, using Appleworks Software by Apple Computer. Subsequently, the design template was transferred with little difficulty to an IBM-PC with a 20MB hard disk and 640K of RAM, using Lotus 1-2-3 software by Lotus Corporation. In general, spreadsheet programs developed for use on microcomputers require at least 128K to 256K of RAM and two disk drives.

TABLE II
System characteristics for each branch duct

A_{slot}	Slot area, m^2 (ft^2)
L_i	Duct length, m (ft)
D_i	Duct diameter, mm (in.)
R/d	Elbow curvature
VP/elbow	Elbow loss factor
Number of elbows (based on fractional elbows)	
Acceleration loss, 1VP/hood	
Slot entry loss	
Hood entry loss	
Branch entry loss	
Other characteristics	
Air cleaner type	
Rated volume flow standard m^3/s (scfm)	
Rated pressure drop mm H_2O (in.H_2O)	

This article contains the equations and correlations necessary to perform ventilation design calculations for standard and nonstandard atmospheric conditions. These relationships, based on perfect gas law thermodynamic properties of moist air, take the place of psychrometric charts used in hand calculations. Once these relations are incorporated into the chosen spreadsheet program, the need to refer to tabulated data is largely eliminated. In this article, the equations are those appropriate to the "velocity pressure method" for ventilation design, in which most pressure losses are described in terms of the duct velocity pressure. Alternatively, one could create a template to perform the calculations using the "equivalent foot method," in which pressure losses in components such as hoods are equated to the pressure loss in an "equivalent length of straight duct."

System design

The primary purpose of ventilation system design is to specify the fan size in terms of volumetric airflow and fan static pressure. Ventilation design requires that system components and some of their operating characteristics be determined *a priori*. The initial step involves the layout of the processes or machines for which the system is being designed. The ventilation system used to illustrate this spreadsheet design process is shown in Figure 1. The appropriate hoods are selected or designed and air cleaner types selected.[4–5] The ductwork configuration and air cleaner and fan locations are specified.

The properties of the working fluid (usually air) are specified along with some operating characteristics such as minimum transport velocity through the ducts. Some design information or system characteristics are to be manually selected and supplied to the program. Table I indicates the operating characteristics; Table II indicates the system characteristics which must be manually selected. These values will become input data to the spreadsheet model.

Characteristics of air and psychrometric calculations

Ventilation pressure loss tables are based on standard conditions of air, i.e., air with a density of 1.2 kg/m³ (0.075 lb/ft³), a temperature of 21°C (70°F), barometric pressure of 760 mm Hg (29.92 in. Hg), and relative humidity of 50%. "Standard air" has a humidity ratio of 0.00796 kg H_2O/kg dry air (0.00796 lbs H_2O/lbs dry air) (55 grains/lbs dry air). If the density of air in any part of the system deviates significantly from the standard conditions, the actual pressure losses through the system will be different than those predicted by calculations based on standard air. In such situations, corrections for the difference in density must be made. Such corrections are easily implemented if the designer knows the dry-bulb temperature and the humidity ratio. The equations given in this paper incorporate the factors for nonstandard air; hence there is no need to account for nonstandard conditions. For standard conditions, the input values listed in Table I would be simply T_{DB} = 21°C (70°F), P_{baro} = 760 mm Hg (29.92 in. Hg) and w = 0.00785 kg H_2O/kg air (lbs H_2O/lbs air).

The dry air volume flow rate ($Q_{dry\ air}$, standard m³/s [SCFM]) for each hood, slot, or entry is specified. The dry air mass flow rate ($m_{dry\ air}$, kg/s [lbs/min]) is determined from the volume flow rate and the dry air density:

$$m_{dry\ air} = (Q_{dry} * 0.07492) \qquad (1)$$

The actual volume flow rate (Q_{actual}, actual m³/s [ACFM]) is computed from the dry air mass flow rate and the humid air volume.

$$Q_{actual} = m_{dry\ air} * v \qquad (2)$$

where:

Q_{actual} = actual volume flow

v = the humid air volume, m³/kg of dry air (ft³/lb of dry air)

The humid air volume may be computed from

$$\dot{v} = K_1 * (T_{DB} + K_2) * (1 + 1.6055\ w)/P_b \qquad (3)$$

where:

T_{DB} = dry-bulb temperature, C (F)

w = humidity ratio, kg H_2O/kg dry air (lbs H_2O/lbs dry air)

P_b = total barometric pressure, mm Hg (in. Hg)

The enthalpy (h, J/kg [Btu/lb of dry air]) of the air may be determined from

$$h = (K_3 * T_{DB}) + (w * (K_4 + (K_5 * T_{DB}))) \qquad (4)$$

The relation for v is derived from the perfect gas equation of state.[6–7] To derive the relation for the mixture enthalpy, a nonreacting mixture of perfect gases is assumed.[6]

At a point in the ventilation system where "standard air" mixes with nonstandard air, (for example, downstream of an evaporative dryer or a humidifier), the dry bulb temperature of the mixed airstream must be computed:[6]

$$T_{DB,mix} = (m_1 * T_{DB1} + m_2 * T_{DB2})/(m_1 + m_2) \qquad (5)$$

where:

$T_{DB,mix}$ = dry-bulb temperature of mixed airstream

m_i = mass flow of each airstream, i = 1,2

T_{DBi} = dry-bulb temperature, C (F) of each airstream

Similarly, the humidity ratio of the mixture can be calculated by substituting w in place of T_{DB} in equation 5. Equations 2 and 3 can then be used to compute the characteristics of the new humid air mixture.

The moist air density and a density correction factor must also be computed.

Moist air density:

$$\rho_{moist\ air} = (1/v) * (1 + w) \qquad (6)$$

Density correction factor:

$$DCF = \rho_{moist\ air} / \rho_{STP} \qquad (7)$$

System components and characteristics

Normally in the course of ventilation design, one computes the pressure drops or losses due to friction and losses due to turbulence and mixing using tabular data or graphs.[4,5,8] When using a spreadsheet to perform the calculations, the normally tabulated data or graphical data are computed directly using the appropriate equations or correlations. These equations are inserted into the spreadsheet in the appropriate cell. For each of the system components, the input parameters are discussed, and the important equations are listed below.

Ducts: For each end point in the system, as well as connecting sections of duct, duct length (L_i) and duct diameter (D_i) must be specified (Table II). From the diameter, the duct cross-sectional area is computed:

$$A_{duct} = ((D_i/2)^2) * \pi/K_6 \qquad (8)$$

The duct velocity is

$$V_{duct} = Q_{actual}/A_{duct} \qquad (9)$$

Duct velocity pressure (duct VP in mm H_2O (in. H_2O)) is computed from the following equation:

$$VP_{duct} = \rho_{moist\ air} * ((V_{duct}/K_7)^2) \qquad (10)$$

The straight duct loss factor, VP per meter (VP per 100 feet), is then computed:

$$VP' = K_8/(D_i^{1.22}) * (VP_{duct}^{0.95})/VP_{duct} \qquad (11)$$

and finally, the straight duct losses in units of velocity pressure are computed:

$$\text{Duct losses} = (L_i * VP')/100 \qquad (12)$$

Slot: If the system has slots present, the slot cross-sectional area must be supplied. The slot velocity is then computed from the slot area and the actual volume flow rate:

$$V_{slot} = Q_{acfm}/A_{slot} \qquad (13)$$

The slot velocity pressure (mm H_2O [in. H_2O]) is then computed

$$VP_{slot} = \rho_{moist\ air} * ((V_{slot}/K_7)^2) \qquad (14)$$

The plenum loss factor is the sum of the acceleration loss and the slot entry loss, both specified to the program at the outset:

$$f_{plenum} = 1\ VP + VP_{slot} \qquad (15)$$

Then the static pressure loss in mm H_2O (in. H_2O) units is determined:

$$SP_{plenum} = f_{plenum} * VP_{slot} \qquad (16)$$

The elbow radius of curvature (radius/duct diameter) and the elbow loss factor (from tables) are initial inputs to the spreadsheet. By multiplying the number of fractional elbows and the loss factor, the elbow losses (in units of VP) are determined:

$$\text{elbow losses} = \text{VP/elbow} * \text{Number of elbows} \qquad (17)$$

Other losses computed in mm H_2O (in. H_2O) units are related to significant duct elevation changes or to an air cleaner or other device in the branch which may cause the fluid to experience a pressure drop. The air cleaner type (or types), its rated volume flow (standard m³/s [scfm]) and its rated pressure drop (in mm H_2O) [in. H_2O]) are specified. The operating pressure drop is then calculated:

$$P = (Q_{dry}/Q_{rated})^2 * P_{rated} * DCF \qquad (18)$$

Equation 18 is appropriate for many types of air cleaners such as cyclones. However, other air cleaners may have pressure drops proportional to Q or Q to some other power. Note that the density correction factor (DCF) is used when the pressure loss is given in units of mm H_2O (in. H_2O) rather than units of velocity pressure.

The loss from an elevation change is given by:

$$z_{loss} = (z * \rho_{moist} * K_9)/\rho_{water} \qquad (19)$$

Other losses can occur when there is a contraction in the duct system or an expansion. In the case of an expansion, the static regain is accounted for as follows:

$$SP_1 - SP_2 = VP_1 * R * (B^4 - 1) \qquad (20)$$

		C	D	E	F	G
2	-DUCT DESIGN SPECIFICATIONS	----	----	----	----	----
3	Duct bran[illegible]er or label	A->C	B->C	C->D	D->E	F->E
4	Dry air vo[illegible]flow, SCFM			([illegible]D4)	(E4)	(F4)
5	Dry air mass flow, lbs/min	(C4*0.07492)	(D4*0.07492)	(E4*0.07492)	(F4*0.07492)	(G4*0.07492)
6	Air pres.@ branch start, "Hg			@MAX(C55..D55)	(E55)	
7	Dry bulb temp, 'F			((C5*C7)+(D5*D7))/E5	(E7)	(F7)
8	Humidity ratio, lbH2O/lb air			((C5*C8)+(D5*D8))/E5	(E8)	(F8)
9	Humid air vol. cuft/lb air	0.754*(C7+459.7)*(1+(1.6055*C8))/C6	0.754*(D7+459.7)*(1+(1.6055*D8))/D6	0.754*(E7+459.7)*(1+(1.6055*E8))/E6	0.754*(F7+459.7)*(1+(1.6055*F8))/F6	0.754*(G7+459.7)*(1+(1.6055*G8))/G6
10	Moist air density, lb/cuft	(1/C9*(1+C8))	(1/D9*(1+D8))	(1/E9*(1+E8))	(1/F9*(1+F8))	(1/G9*(1+G8))
11	Density correction factor	(C10/0.07492)	(D10/0.07492)	(E10/0.07492)	(F10/0.07492)	(G10/0.07492)
12	Enthalpy, BTU/lb air	0.24*C7+(C8*(1061.2+(0.45*C7)))	0.24*D7+(D8*(1061.2+(0.45*D7)))	0.24*E7+(E8*(1061.2+(0.45*E7)))	0.24*F7+(F8*(1061.2+(0.45*F7)))	0.24*G7+(G8*(1061.2+(0.45*G7)))
13	Actual volume flow rate, ACFM	(C9*C5)	(D9*D5)	(E9*E5)	(F9*F5)	(G9*G5)
14	slot area, sqft		@NA	@NA	@NA	@NA
15	slot velocity, fpm	(C13/C14)	(D13/D14)	(E13/E14)	(F13/F14)	(G13/G14)
16	VP slot, "H2O	((C15/1096.8144)Δ2)*C10	((D15/1096.8144)Δ2)*D10	((E15/1096.8144)Δ2)*E10	((F15/1096.8144)Δ2)*F10	((G15/1096.8144)Δ2)*G10
17	Acceleration loss, 1 VP					
18	Slot entry loss, VP slot					
19	Plenum loss factor, VP	@SUM(C17..C18)	@SUM(D17..D18)	@SUM(E17..E18)	@SUM(F17..F18)	@SUM(G17..G18)
20	Plenum SP, "H2O	(C16*C19)				
21	Duct length, ft					
22	Duct dia. inches					
23	Duct area, sqft	(((C22/2)Δ2)*3.14159)/144	(((D22/2)Δ2)*3.14159)/144	(((E22/2)Δ2)*3.14159)/144	(((F22/2)Δ2)*3.14159)/144	(((G22/2)Δ2)*3.14159)/144
24	Duct Velocity, fpm	(C13/C23)	(D13/D23)	(E13/E23)	(F13/F23)	(G13/G23)
25	Duct VP, "H2O	((C24/1096.8144)Δ2)*C10	((D24/1096.8144)Δ2)*D10	((E24/1096.8144)Δ2)*E10	((F24/1096.8144)Δ2)*F10	((G24/1096.8144)Δ2)*G10
26	St. Duct friction, VP/100ft	38.265/(C22Δ1.22)*(C25Δ0.95)/C25	38.265/(D22Δ1.22)*(D25Δ0.95)/D25	38.265/(E22Δ1.22)*(E25Δ0.95)/E25	38.265/(F22Δ1.22)*(F25Δ0.95)/F25	38.265/(G22Δ1.22)*(G25Δ0.95)/G25
27	Elbow curvature, R/dia					
28	Elbow loss factor, VP elbow					
29	Number of elbows in branch					
30	-BRANCH LOSSES IN VP UNITS	----	----	----	----	----
31	Acceleration loss, 1 VP hood					
32	Hood entry loss, VP hood					
33	Straight duct loss, VP	(C21*C26)/100	(D21*D26)/100	(E21*E26)/100	(F21*F26)/100	(G21*G26)/100
34	Elbow loss, VP	(C28*C29)	(D28*D29)	(E28*E29)	(F28*F29)	(G28*G29)
35	Branch entry loss factor, VP					
36	Other losses in VP units					
37	Subtotal, lines 31 to 36	@SUM(C31..C36)	@SUM(D31..D36)	@SUM(E31..E36)	@SUM(F31..F36)	@SUM(G31..G36)
38	-OTHER LOSSES IN "H2O UNITS	----	----	----	----	----
39	Elevation change, ft					
40	Air cleaner type			36"CYCL.		
41	Rated volume flow, acfm					
42	Rated pressure drop, "H2O					
43	Operating volume flow, acfm			(E13)		
44	Operating pres. drop, "H2O			(E42*E11*(E13/E41)Δ2)		
45	Loss from elev. change, "H2O	(C39*C10*12)/62.303	(D39*D10*12)/62.303	(E39*E10*12)/62.303	(F39*F10*12)/62.303	(G39*G10*12)/62.303
46	Slot/Plenum loss, "H2O line 20	(C20)	(D20)	(E20)	(F20)	(G20)
47	Other losses in "H2O units					
48	Subtotal, lines 44 to 47	@SUM(C44..C47)	@SUM(D44..D47)	@SUM(E44..E47)	@SUM(F44..F47)	@SUM(G44..G47)
49	-SUMMARY OF LOSSES, "H2O	----	----	----	----	----
50	VP losses, line 37 x VP ("H2O)	(C25*C37)	(D25*D37)	(E25*E37)	(F25*F37)	(G25*G37)
51	Losses in "H2O units, line 48	(C48)	(D48)	(E48)	(F48)	(G48)
52	Total branch loss, line 50+51	@SUM(C50..C51)	@SUM(D50..D51)	@SUM(E50..E51)	@SUM(F50..F51)	@SUM(G50..G51)
53	DESIGN SUMMARY	----	----	----	----	----
54	Cumulative loss @ branch end	(C52)	(D52)	@MAX(C54..D54)+E52	(E54+F52)	(F54+G52)
55	Static pres.@ branch end, "Hg	(C6-(C54/13.6025))	(D6-(D52/13.6025))	(E6-(E52/13.6025))	(F6-(F52/13.6025))	(G6+(G52/13.6025))
56	Fan total pres. @ NTP, "H2O	----	----	----	----	(F54+G52)/F11
57	Fan static pres. @ NTP, "H2O	----	----	----	----	(F54+G52-F25)/F11
58	Fan brake horsepower @60% eff.	----	----	----	----	(G56*G13)/(6356*0.6)

Figure 2—Ventilation design spreadsheet template.

where:

B = ratio of upstream diameter to downstream diameter

R = regain factor

SP_1 = static pressure, upstream of expansion

SP_2 = static pressure, downstream of expansion

In the case of a contraction, the change in static pressure is computed from

$$SP_1 - SP_2 = VP_1 * (B^4 - 1) * (1 + L) \quad (21)$$

where:

L = loss factor

The loss factor and the regain factors are determined from tables or charts in standard ventilation references.[4,8]

As in the traditional worksheet, one begins with each branch, calculating the pressure loss through the hood, duct and other components. The total loss in the branch is then computed. It is easiest to subtotal separately the losses computed in terms of velocity pressure and those computed in terms of mm H_2O (in. H_2O). The losses computed in terms of velocity pressure are then converted to mm H_2O (in. H_2O) using equation 22:

$$SP\ losses = VP_{duct} * Subtotal\ of\ VP\ losses \quad (22)$$

where:

VP_{duct} = mm H_2O (in. H_2O)

The cumulative static pressure losses in the branch (read as gauge pressures) are then summed, and the absolute pressure (mm Hg [in. Hg]) at the branch end is then computed:

$$ASP_{branch\ end} = ASP_{branch\ start} - (SP_{cum\ loss}/13.6) \quad (23)$$

The absolute static pressure at the branch end ($ASP_{branch\ end}$) is used as the input pressure for the subsequent branch, thus accounting for density changes due to the pressure losses in the system.

When the branch duct reaches a junction with another duct, the cumulative static pressure losses (in mm H_2O [in. H_2O]) must be nearly equal (within 5%) in each duct which meets at the junction. For a design that does not use mechanical dampers to achieve the proper air distribution, the static pressure balancing method is used. The percent difference in the static pressures between the adjoining ducts is determined. Normal convention refers to the duct with the greater cumulative pressure loss as the governing duct at a junction. If the cumulative pressure losses differ by less than 5 percent, no recalculation is necessary. If the difference is between 5 percent and 20 percent, balancing may be achieved by increasing the airflow through the non-governing duct with equation 24. If the difference is greater than 20 percent, one of the ducts must be modified by changing the diameter, length, etc.

$$Q_{adjusted} = Q_{design}\ [higher\ SP_{loss}/lower\ SP_{loss}]^{0.5} \quad (24)$$

where:

Q_{adjust} = corrected volume flow rate through the non-governing duct

Q_{design} = original flow rate in the non-governing duct

SP = static pressure

It is standard practice to use the governing duct pressure at a junction when calculating the cumulative losses in subsequent duct segments. This conservative approach insures the airflow in all duct segments will be at least as large as specified in the design. (The @MAX function can be used in lines 6 and 55 of Figure 2 to automatically select the governing static pressure.)

The static pressure calculated at a junction with the above procedure is sufficient if the velocity pressure following the junction does not exceed the velocity pressure in the incoming branches. Generally, small differences in velocity pressure below 0.1 inch H_2O may be ignored. Otherwise, the static pressure at the junction must be increased to account for the additional acceleration of the air (see ref. 4, pp. 6–8). (This correction would be included in line 36 of Figure 2.)

At those junctions where a branch duct joins the main duct, the total mass flow of dry air (at STP) at that junction is computed. By working with mass flow rates, a constant dry air mass flow rate is assured in all parts of the system, and, hence, a mass balance is assured. Values are then converted as before to actual volume flow rates which are a function of the temperature and humidity at that point in the system.

Because the pressure in the exhaust stack is positive (i.e., above atmospheric pressure), and pressure losses in the stack must be included to size the fan correctly, the direction of calculation of stack losses must be carried out from the stack outlet back to the fan outlet.

The fan characteristics are computed using the following relationships:

$$Fan\ TP = (Cum\ losses\ at\ fan\ outlet + stack\ losses)/DCF \quad (25)$$

TABLE III
Conversion constants for English and metric units

Constants	English engineering	MKS metric
K_1	0.7544	287.06
K_2	459.7	273.15
K_3	0.24	1.0
K_4	1064.2	2501.3
K_5	0.45	1.86
K_6	144	10^6
K_7	1096.253	4.4285
K_8	38.256	78.62
K_9	12	10^3
K_{10}	6356	9.8
ρ_{water}	62.303	10^3
ρ_{STP}	0.07492	1.2

$$Fan\ SP = (SP_{cum\ loss\ fan\ inlet} + SP_{stack\ loss} - VP_{duct\ fan\ inlet})/DCF \quad (26)$$

$$Fan\ BHP = (Fan\ TP * Q_{actual\ total})/(K_{10} * 0.6) \quad (27)^*$$

where 0.6 is the assumed mechanical efficiency of the fan. Equations 25–27 may be used to select from standard air rating tables a fan that will operate at nonstandard conditions. Under certain nonstandard conditions, provision for cold startup must be made.[4]

Spreadsheet structure and program development

To begin the design process using a spreadsheet program, one must first construct the spreadsheet template using the format appropriate for the software of the designer's choice. The spreadsheet is a matrix: each intersection of a row and a column defines a cell, identified by its row and column. For the ventilation design, each column represents a branch duct (e.g., beginning with a hood and ending with a junction with another duct), a section of main duct, or the exhaust stack. Each row represents a system component characteristic, an operating condition, or a computed value. In general, one inserts the row and column labels first and then constructs the template from the top down. When constructing the model, the equation for each computed value is inserted into the correct cell of the spreadsheet. For example, cell E23 (column E, row 23) contains the equation for computing the duct area in branch C-D (Figure 2).

To use Equations 1–27 above, one simply changes the name of each variable in the equation to the correct row-column

*Power in watts for metric system

A 1	B	E	
2	-DUCT DESIGN SPECIFICATIONS	------------------------------------	
3	Duct branch number or label	C->D	
4	Dry air volume flow, SCFM	(C4+D4)	Table Ia & Fig. 1
5	Dry air mass flow, lbs/min	(E4*0.07492)	Equation 1
6	Abs. pres.@ branch start, "Hg	@MAX(C55..D55)	Table Ia & Equation 23
7	Dry bulb temp, 'F	((C5*C7)+(D5*D7))/E5	Table Ia & Equation 5
8	Humidity ratio, lbH2O/lb air	((C5*C8)+(D5*D8))/E5	Table Ia & Equation 5
9	Humid air vol. cuft/lb air	0.754*(E7+459.7)*(1+(1.6055*E8))/E6	Equation 3
10	Moist air density, lb/cuft	(1/E9*(1+E8))	Equation 6
11	Density correction factor	(E10/0.07492)	Equation 7
12	Enthalpy, BTU/lb air	0.24*E7+(E8*(1061.2+(0.45*E7)))	Equation 4
13	Actual volume flow rate, ACFM	(E9*E5)	Equation 2
14	slot area, sqft	@NA	
15	slot velocity, fpm	(E13/E14)	Equation 13
16	VP slot, "H2O	((E15/1096.8144)Δ2)*E10	Equation 14
17	Acceleration loss, 1 VP		Table 1b & Fig. 1
18	Slot entry loss, VP slot		Table 1b & Fig. 1
19	Plenum loss factor, VP	@SUM(E17..E18)	Equation 15
20	Plenum SP, "H2O		Equation 16
21	Duct length, ft		Table 1b & Fig. 1
22	Duct dia. inches		Table 1b
23	Duct area, sqft	(((E22/2)Δ2)*3.14159)/144	Equation 8
24	Duct Velocity, fpm	(E13/E23)	Equation 9
25	Duct VP, "H2O	((E24/1096.8144)Δ2)*E10	Equation 10
26	St. Duct friction, VP/100ft	38.265/(E22Δ1.22)*(E25Δ0.95)/E25	Equation 11
27	Elbow curvature, R/dia		Table 1b & Fig. 1
28	Elbow loss factor, VP elbow		Table 1b
29	Number of elbows in branch		Table 1b & Fig. 1
30	-BRANCH LOSSES IN VP UNITS	------------------------------------	
31	Acceleration loss, 1 VP hood		Table 1b
32	Hood entry loss, VP hood		Table 1b & Fig. 1
33	Straight duct loss, VP	(E21*E26)/100	Equation 12
34	Elbow loss, VP	(E28*E29)	Equation 17
35	Branch entry loss factor, VP		Table 1b
36	Other losses in VP units		Table 1b
37	Subtotal, lines 31 to 36	@SUM(E31..E36)	
38	-OTHER LOSSES IN "H2O UNITS	------------------------------------	
39	Elevation change, ft		Table 1b & Fig. 1
40	Air cleaner type		Table 1b & Fig. 1
41	Rated volume flow, acfm		Table 1b & Fig. 1
42	Rated pressure drop, "H2O		Table 1b & Fig. 1
43	Operating volume flow, acfm	(E13)	Equation 2
44	Operating pres. drop, "H2O	(E42*E11*(E13/E41)Δ2)	Equation 18
45	Loss from elev. change, "H2O	(E39*E10*12)/62.303	Equation 19
46	Slot/Plenum loss, "H2O line 20	(E20)	Equation 16
47	Other losses in "H2O units		
48	Subtotal, lines 44 to 47	@SUM(E44..E47)	
49	-SUMMARY OF LOSSES, "H2O	------------------------------------	
50	VP losses, line 37 x VP ("H2O)	(E25*E37)	Equation 22
51	Losses in "H2O units, line 48	(E48)	
52	Total branch loss, line 50+51	@SUM(E50..E51)	
53	-DESIGN SUMMARY	------------------------------------	
54	Cumulative loss @ branch end	@MAX(C54..D54)+E52	
55	Abs. pres.@ branch end, "Hg	(E6-(E52/13.6025))	Equation 23
56	Fan total pres. @ NTP, "H2O	------------------------------------	Equation 25
57	Fan static pres. @ NTP, "H2O	------------------------------------	Equation 26
58	Fan brake horsepower @60% eff.	------------------------------------	Equation 27

Figure 3—Example of how equations are translated into spreadsheet formulae.

A 1	B	C	D	E	F	G
2	-DUCT DESIGN SPECIFICATIONS	----------	----------	----------	----------	----------
3	Duct branch number or label	A->C	B->C	C->D	D->E	F->E
4	Dry air volume flow, SCFM	1020.00	2250.000	3270.00	3270.00	3270.00
5	Dry air mass flow, lbs/min	76.418	168.570	244.988	244.988	244.988
6	Air pres.@ branch start, "Hg	29.921	29.921	29.832	29.673	29.921
7	Dry bulb temp, 'F	70.000	400.000	297.064	297.064	297.064
8	Humidity ratio, lbH2O/lb air	0.008	0.040	0.030	0.030	0.030
9	Humid air vol. cuft/lb air	13.520	23.055	20.049	20.156	19.989
10	Moist air density, lb/cuft	0.075	0.045	0.051	0.051	0.052
11	Density correction factor	0.995	0.602	0.686	0.682	0.688
12	Enthalpy, BTU/lb air	25.542	145.648	107.164	107.164	107.164
13	Actual volume flow rate, ACFM	1033.16	3886.457	4911.75	4938.06	4897.15
14	slot area, sqft	1.000	NA	NA	NA	NA
15	slot velocity, fpm	1033.156	NA	NA	NA	NA
16	VP slot, "H2O	0.066	NA	NA	NA	NA
17	Acceleration loss, 1 VP	1.000				
18	Slot entry loss, VP/slot	1.780				
19	Plenum loss factor, VP	2.780	0.000	0.000	0.000	0.000
20	Plenum SP, "H2O	0.184				
21	Duct length, ft	15.000	55.000	25.000	75.000	30.000
22	Duct dia. inches	8.000	14.000	16.000	16.000	16.000
23	Duct area, sqft	0.349	1.069	1.396	1.396	1.396
24	Duct Velocity, fpm	2959.775	3635.555	3517.782	3536.629	3507.327
25	Duct VP, "H2O	0.543	0.496	0.528	0.531	0.527
26	St. Duct friction, VP/100ft	3.121	1.584	1.342	1.341	1.342
27	Elbow curvature, R/dia		2.000			
28	Elbow loss factor, VP/elbow		0.270			
29	Number of elbows in branch		1.000			
30	-BRANCH LOSSES IN VP UNITS	----------	----------	----------	----------	----------
31	Acceleration loss, 1 VP/hood	1.000	1.000			
32	Hood entry loss, VP/hood	0.250	0.290			
33	Straight duct loss, VP	0.468	0.871	0.335	1.006	0.403
34	Elbow loss, VP	0.000	0.270	0.000	0.000	0.000
35	Branch entry loss factor, VP	0.180				
36	Other losses in VP units					
37	Subtotal, lines 31 to 36	1.898	2.431	0.335	1.006	0.403
38	-OTHER LOSSES IN "H2O UNITS	----------	----------	----------	----------	----------
39	Elevation change, ft	0	0	0	0	30
40	Air cleaner type			36"CYCL.		
41	Rated volume flow, scfm			5000.000		
42	Rated pressure drop, "H2O			3.000		
43	Operating volume flow, acfm			4911.747		
44	Operating pres. drop, "H2O			1.985		
45	Loss from elev. change, "H2O	0.000	0.000	0.000	0.000	0.298
46	Slot/Plenum loss, "H2O line 20	0.184	0.000	0.000	0.000	0.000
47	Other losses in "H2O units	0.000				
48	Subtotal, lines 44 to 47	0.184	0.000	1.985	0.000	0.298
49	-SUMMARY OF LOSSES, "H2O	----------	----------	----------	----------	----------
50	VP losses, line 37 x VP ("H2O)	1.031	1.205	0.177	0.534	0.212
51	Losses in "H2O units, line 48	0.184	0.000	1.985	0.000	0.298
52	Total branch loss, line 50+51	1.214	1.205	2.162	0.534	0.510
53	DESIGN SUMMARY	----------	----------	----------	----------	----------
54	Cumulative loss @ branch end	1.214	1.205	3.372	3.907	4.416
55	Static pres.@ branch end, "Hg	29.832	29.832	29.673	29.634	29.958
56	Fan total pres. @ NTP, "H2O	----------	----------	----------	----------	6.475
57	Fan static pres. @ NTP, "H2O	----------	----------	----------	----------	5.696
58	Fan brake horsepower @60% eff.	----------	----------	----------	----------	8.315

Figure 4—Ventilation design spreadsheet.



Figure 5—Results of spreadsheet validation showing predicted values of textbook examples versus ideal values.

label and inserts the equation in the appropriate cell. Appropriate constants (K_1–K_{10}) must be inserted into the above equations from Table III. The choice of constants depends on your decision to use the English engineering or metric system of units. The formulas and values shown in Figures 2–6 are in English units. For example, Equation 8

$$A_{duct} = ((D_i/2)^2) * \pi/K_6 \quad \textbf{(8)}$$

becomes

$$((E22/2)^2) * (3.142/144) \quad \textbf{(28)}$$

where E22 is defined as the duct diameter in inches. There is no need to include the "A_{duct} = " as the spreadsheet assumes that the result of Equation 28 displayed in cell E23 is the area of the duct. Figure 3 illustrates this concept; one representative column and each row of the spreadsheet are shown. The cells are shown with the original equations discussed above and with the actual equations inserted into the spreadsheet. In most spreadsheet programs, once a column is complete, it can be easily copied to other columns such that the structure of the completed spreadsheet will be similar to that shown in Figure 2. Modifications, such as the one in Column E for the cyclone, are easily achieved.

Once the underlying template is completed, the requisite input values for each branch (Tables I and II) are inserted into the appropriate cells. The spreadsheet uses these data in the various formulas to compute the intermediate values in the cells and to complete the worksheet. Figure 4 shows the actual spreadsheet results obtained for the system considered in this paper. It is this table of numbers that one observes on the screen and that one prints out for the results. The "bottom line" is the values G49 to G51 which give the fan static pressure, the fan total pressure, and the brake horsepower at the specified efficiency.

Validation of spreadsheet model

An important step in developing a spreadsheet model is to verify the accuracy of the calculated results. Generally, this auditing process involves supplying the input data for which correct answers are known and then comparing the calculated results to the known answers. In the case of this ventilation design template, two validation steps were carried out.

First, researchers compared hand calculated results given in textbook examples to the results calculated using the spreadsheet template with the same design specifications.[4,8] These examples included calculations for nonstandard conditions, multiple branches, and a variety of duct sizes and hoods.

Second, they measured duct static pressures in several test systems and compared these measurements to predicted static pressures generated by the spreadsheet model. The test systems consisted of different hood and duct sections connected to a venturi meter that measured the mass flow of air through the system.

The results of the first validation test are displayed as a scatterplot of the ideal value from the textbook example (abscissa) against the value predicted by the spreadsheet model (ordinate) (Figure 5). As expected, these points all fall on a diagonal line with a unit slope. A least squares regression line ($y = 1.0x$) fitted to these data yields on $R^2 \geq 0.999$ for each plot. The average error in the predicted values was about −0.5 percent with a maximum range of ±5 percent. A similar plot (Figure 6) shows the spreadsheet model predictions of actual duct static pressures. Here, the average error was +4 percent with a maximum range of ±20 percent. Together these test results indicate the spreadsheet calculations essentially duplicate the text-



Figure 6—Results of spreadsheet validation showing predicted duct static pressures versus observed duct static pressures.

book examples and predict actual duct pressures with a fair degree of accuracy.

Conclusion

Once the design template is complete, the beauty of the spreadsheet approach becomes apparent. For example, changing the duct diameters in cells E22 to G22 from 406 mm (16 in.) to 508 mm (20 in.) results in a savings in fan horsepower of 1.2 HP. In other words, the opportunity for design optimization is readily available.

Finally, spreadsheets are adaptable to a wide variety of other numerical analysis problems such as computing a particle size distribution. Thus, knowledge of a spreadsheet program can provide an industrial hygienist or other professional with a broad-based numerical analysis tool to solve the wide variety of problems often encountered.

Recommendations

Some spreadsheet programs are somewhat inflexible with regard to the sequence in which cells are recalculated. This worksheet is arranged so that calculations proceed from the upper left corner, downward by row and across by column. This is the most widely used calculation method. If users rearrange the worksheet, it may become necessary to recalculate the values several times to get the correct results.

In developing this ventilation template, this study has attempted to keep it general by using formulas and features that can be readily applied to any spreadsheet software program. However, many spreadsheets offer additional capabilities. For example, one can create data look-up tables that automatically insert tabular data, like elbow loss factors, into the appropriate cells when the user supplies the elbow curvature. The availability of built-in functions such as summation {@sum in Figure 4} can simplify the development of the spreadsheet. In Figure 2, for example, the summation function is used to sum the branch losses in units of VP. The more sophisticated programs include functions such as line graphs, logical operators, and programmable macros. The latter could be utilized to implement an automatic duct branch balancing system.

Acknowledgments

Research described in this article was supported by a University of California Regents Instructional Grant.

APPENDIX

List of symbols

A_{duct}	duct cross-sectional area
A_{slot}	slot cross-sectional area
ASP	absolute static pressure
B	diameter ratio
D_i	duct diameter
DCF	density correction factor
f_{plenum}	plenum loss factor
Fan BHP	fan brake horsepower
Fan SP	fan static pressure
Fan TP	fan total pressure
h	enthalpy (BTU/lb of dry air)
K_{1-10}	conversion constants in table 2
L_i	duct length
$m_{dry\ air}$	dry air mass flow rate (lbs/min)
m_i	mass flow of each airstream (i = 1,2)
P_b	total barometric pressure (in. Hg)
P_{rated}	rated pressure drop of air cleaner (in. H_2O)
Q_{actual}	actual volume flow rate of moist air
Q_{adjust}	corrected volume flow rate in duct with lower SP
Q_{design}	original volume flow rate in duct with lower SP
Q_{dry}	dry air volume flow rate (scfm)
Q_{rated}	rated volume flow rate of air cleaner (scfm)
R	regain factor
SP	static pressure (in. Hg)
$SP_{cum\ loss}$	cumulative static pressure loss at branch end
SP_{plenum}	plenum static pressure
T_{DB}	dry bulb temperature
$T_{DB,mix}$	dry bulb temperature of mixed airstream
v	humid air volume (ft^3/lb dry air)
V_{duct}	duct velocity
VP_{duct}	duct velocity pressure
VP_{elbow}	elbow loss factor
VP_{slot}	slot velocity pressure
VP′	straight duct loss factor (VP/100 feet of duct)
w	humidity ratio (lbs H_2O/lbs dry air)
z	elevation
Δ_z	elevation change
ρ_{moist}	density of moist air
ρ_{stp}	density of (dry) standard air
ρ_{water}	density of water @ STP

Note: A * indicates multiplication, and a Δ indicates exponentiation in Figures 2 and 3.

References

1. Clapp, D.E.; Groh, D.S.; Nenadic, J.D.: Ventilation Design by Microcomputer. Am. Ind. Hyg. Assoc. J. 43:212 (1982).
2. Loeffler, J.J.: Using Programmable Calculators for Ventilation System Design. II. Plant Engineering, p. 49 (Sept. 14, 1978).
3. Shotwell, H.P.: A Ventilation Design Program for Hand-Held Programmable Computers. Am. Ind. Hyg. Assoc. J. 45:749 (1984).
4. Industrial Ventilation, 19th ed. Am. Conf. Govt. Ind. Hyg., Cincinnati, OH (1986).
5. McDermott, H.J.: Handbook of Ventilation for Contaminant Control, 2nd ed., pp. 1–402. Butterworth Publishers, Boston (1985).
6. Clifford, G.E.: Heating, Ventilating and Air Conditioning. Reston Publishing Co., Inc., Reston, VA (1984).
7. Science and Engineering Programs, Apple II Edition, pp. 192–194. J. Heilborn, Ed. Osborne/McGraw-Hill, Berkeley (1981).
8. ASHRAE Handbook: 1985 Fundamentals. American Society of Heating, Refrigeration and Air-Conditioning Engineers, Atlanta, GA (1985).
9. Burton, D.J.: Industrial Ventilation, pp. 1–213. IVE, Inc., Salt Lake City, UT (1982).

Received 2/10/87; review/decision 3/23/87; revision 4/21/87; accepted 4/27/87